

EuroOffice Map Chart Demonstration

The MapData function is used to collect the total values per country.

LEFT([MapData]; LEN([MapData]) - 2) is then used as Data Source for the Map Chart.

The MapData function is a bit complicated because it has to work around a few bugs in Pentaho Reporting:

```
IF( FIND( [Location]; [MapData] & [Location] ) = LEN( [MapData] ) + 1
;
  [MapData] & [Location] & ": " & TEXT( [Value] ) & ", "
;
  SUBSTITUTE(
    [MapData];
    [Location] & ": " & LEFT(
      RIGHT(
        [MapData];
        LEN( [MapData] ) - (FIND( [Location]; [MapData] ) + LEN( [Location] ) + 2) + 1
      );
    FIND( ", "; [MapData]; FIND( [Location]; [MapData] ) + LEN( [Location] ) + 2 ) -
    (FIND( [Location]; [MapData] ) + LEN( [Location] ) + 2)
  ) & ", ";
  [Location] & ": " & TEXT( [Value] + INT( LEFT(
    RIGHT(
      [MapData];
      LEN( [MapData] ) - (FIND( [Location]; [MapData] ) + LEN( [Location] ) + 2) + 1
    );
    FIND( ", "; [MapData]; FIND( [Location]; [MapData] ) + LEN( [Location] ) + 2 ) -
    (FIND( [Location]; [MapData] ) + LEN( [Location] ) + 2)
  ))) & ", "
)
)
```

In the end it creates a string like "Belgium: 12, France: 15, Germany: 24, Hungary: 27, ".
 (The LEFT() function is used in the Data Source definition to get rid of the last comma.)

Feel free to use this function in your own reports.

Location	Belgium
Total Value	12
Location	France
Total Value	15
Location	Germany
Total Value	24
Location	Hungary
Total Value	27

